



ABSTRACT

The macro facility in the SAS System gives you extra functionality, more programming tools and a high level language for controlling Data Steps and Proc Steps. With the macro facility, you can pass values from step to step in normal SAS code, you can create bundles of SAS code to automate your work, and you can set up mechanisms for conditionally executing your SAS code. This paper introduces the macro facility and provides some basic SAS macro examples.

INTRODUCTION

Within a SAS program, all the SAS Data Steps and Proc Steps are compiled and executed separately. This structure allows great flexibility but imposes some restrictions on the programmer:

- How do we pass values from step to step?
- How do we use a value that might change from day to day, for example the date?
- How do we execute a Proc Step that depends on the value found in a previous Data Step?
- How do we execute a Data Step or Proc Step a number of times, dependant on the value of a variable?
- How can we construct code that depends on the value of a variable?

The macro facility provides the answer to all these problems. Using the macro facility can also make your programs more generic and flexible, making the code you write more usable and powerful.

THE SAS MACRO FACILITY

The SAS macro facility consists of 3 areas:

1. The ability to create "macro variables", and use automatic "macro variables" (those that exist already)
2. The ability to create "macro code bundles"
3. The ability to use the "macro programming language"

The macro facility has 2 "triggers":

1. The & sign followed by a non-blank character
2. The % sign followed by a non-blank character

When a section of SAS code is processed, the occurrence of a macro "trigger" causes the SAS supervisor to pass the section to the macro facility which in turn attempts to "resolve" it. Resolving a macro reference means finding the value it takes and giving that value back to the supervisor.

CREATING AND USING MACRO VARIABLES

One way to create a macro variable is to use a %LET statement:

```
%let mvar=cvalue;
```

We have created a macro variable called mvar with a value of cvalue.

- There are no quotation marks around the value
- The macro variable is created as mvar but referred to as &mvar
- leading and trailing blanks are ignored.

Let's create and use a macro variable in some SAS code:

```
PROGRAM EDITOR - (Untitled)
%let dsetname=tuesday;
data &dsetname;
  set sasuser.demog;
run;
```



```
LOG - (Untitled)
1
2   %let dsetname=tuesday;
3
4   data &dsetname;
5       set sasuser.demog;
6   run;

NOTE: The data set WORK.TUESDAY has 104 observations and 15 variables.
NOTE: The DATA statement used 0.82 seconds.
```

Notice the SAS code in the Log does not show the real name of the dataset.

Automatic Macro Variables

These are macro variables that are created by the SAS System at start-up.

If you are using a later release of the SAS System, for example, Release 6.11 or Release 6.12, you can use the following program to examine the automatic macro variables and their values:

```
PROGRAM EDITOR - (Untitled)
options nocenter nodate nonumber;

title "Automatic Macro Variables and Their Values";


proc print data=sashelp.vmacro noobs label uniform;
    var name value;
    format name $15. value $50.;
    label name='Automatic Macro Variable Name'
           value='Value';
    where upcase(scope)='AUTOMATIC';
run;
```

```
OUTPUT - (Untitled)
Automatic Macro Variables and Their Values

Automatic Macro
Variable Name      Value
AFDSID             0
AFDSNAME
AFLIB
AFSTR1
AFSTR2
FSPBDV
SYSBUFFR
SYSCMD
SYSDATE            23JUN97
SYSDAY             Monday
SYSDEVIC
SYSDSN             WORK    TUESDAY
SYSENV             FORE
SYSERR             0
SYSFILRC           0
SYSINDEX           0
SYSINFO            0
SYSJOBID           4294819139
SYSLAST            WORK.TUESDAY
SYSLCKRC           0
SYSLIBRC           0
SYSTEMV
SYSMSG
SYSPARM
SYSPBUFF
SYSRC              0
```

Alternatively, you can use the following %PUT statements to report on:


1. The automatic macro variables
2. Those created by the programmer outside a macro code bundle
3. Those created by the programmer inside a macro code bundle



```
PROGRAM EDITOR - (Untitled)
%put _automatic_;
%put _global_;
%put _local_;
```


The values are placed in the Log, appearing before the relevant statement:

The %put _automatic_ statement creates similar output to that on the previous page and is not shown here.



```
LOG - (Untitled)
62
63 %put _automatic_;
GLOBAL DSETNAME tuesday
64
65 %put _global_;
66 %put _local_;
67
68
69
70
```

We could use the automatic macro variable &sysdate in a SAS program as follows:



```
PROGRAM EDITOR - (Untitled)
proc print data=sasuser.demog noobs uniform;
    var age gender salary;
    title "Report Generated on &sysdate";
run;
```

OUTPUT - (Untitled)

Report Generated on 23JUN97

AGE	GENDER	SALARY
31	M	25200.73
36	M	19200.48
43	M	24700.90
47	M	31900.92
33	M	21500.34
46	M	26600.24
47	M	29200.18
52	M	35200.15
23	M	18600.95
64	M	41900.54

Notice the use of the double quotation marks around the title text.
The following report shows the effect of using single quotation marks:

OUTPUT - (Untitled)

Report Generated on &sysdate

AGE	GENDER	SALARY
31	M	25200.73
36	M	19200.48
43	M	24700.90
47	M	31900.92
33	M	21500.34
46	M	26600.24
47	M	29200.18
52	M	35200.15
23	M	18600.95
64	M	41900.54

A Possible Problem

One possible problem in using macro variables is illustrated by the following program:

```
PROGRAM EDITOR - (Untitled)
%let dsetname=tuesday;
%let library=sasuser;
%let readfile=demog;

data &dsetname;
    set &library.&readfile;
run;
```

The SAS log from this program shows an error:

```
LOG - (Untitled)
90 %let dsetname=tuesday;
91 %let library=sasuser;
92 %let readfile=demog;
93
94 data &dsetname;
95     set &library.&readfile;
NOTE: Line generated by the macro variable "READFILE".
95     sasuserdemog
-----
      307
96 run;

ERROR 307-185: The data set name cannot have more than 8 characters.

NOTE: The SAS System stopped processing this step because of errors.
WARNING: The data set WORK.TUESDAY may be incomplete. When this step was
WARNING: Data set WORK.TUESDAY was not replaced because this step was
NOTE: The DATA statement used 0.42 seconds.
```

A full stop after a macro variable name can act as a delimiter to the name itself. We use 2 full stops in the program, one to act as the end of macro variable name marker, and one for the full stop between the library and dataset name:

```
PROGRAM EDITOR - (Untitled)

%let dsetname=tuesday;
%let library=sasuser;
%let readfile=demog;

data &dsetname;
    set &library..&readfile;
run;
```

The SAS Log now reports no errors:

```
LOG - (Untitled)
98 %let dsetname=tuesday;
99 %let library=sasuser;
100 %let readfile=demog;
101
102 data &dsetname;
103     set &library..&readfile;
104 run;

NOTE: The data set WORK.TUESDAY has 104 observations and 15 variables.
NOTE: The DATA statement used 0.71 seconds.
```

CREATING MACRO CODE BUNDLES

A Macro code bundle is commonly referred to as “a macro”, as distinct from a “macro variable” which we have just been discussing.

A macro code bundle allows you to define what the bundle will contain, and then run it.

Define the macro bundle using %macro and %mend statements:

```
PROGRAM EDITOR - (Untitled)
%macro print;
  proc print data=sasuser.demog noobs uniform;
    var age gender salary;
    where age>45;
    title "Senior Salaries for Employees Over 45 years";
  run;
%mend print;
```

The macro is stored in compiled form in the work.sasmacr catalog and is ready to be run.

Run the macro by "calling" it :

```
PROGRAM EDITOR - (Untitled)
%pr int
```

```
OUTPUT - (Untitled)
Senior Salaries for Employees Over 45 years
```

AGE	GENDER	SALARY
56	F	47520.38
52	F	23760.44
60	F	20592.16
56	F	13760.27
46	F	47520.92
65	F	13840.04
48	F	15840.38
57	F	33520.83
51	F	22760.99
61	F	22592.02
55	F	14840.93
61	F	25410.14

Notice the macro call requires no semi-colons, as the code that it generates contains all the semi-colons that are required.


Passing Parameters

We can also create a macro code bundle that expects to be passed several parameters.

```
PROGRAM EDITOR - (Untitled)
%macro print(dsetname,var1,var2,var3,options,agevalue);
  proc print data=&dsetname &options;
    var &var1 &var2 &var3;
    where age>&agevalue;
    title "Salaries for Employees Over &agevalue years";
  run;
%mend print;

%pr int(sasuser.demog,gender,status,salary,n noobs,35)
```


Notice that we have used &agevalue in 2 places in the code bundle, and also that the &options macro variable can contain as many Proc Print options as we require.



OUTPUT - (Untitled)

Salaries for Employees Over 35 years

GENDER	STATUS	SALARY
F	M	23760.44
F	M	9870.34
F	M	22760.99
F	M	13840.86
F	M	14840.27
F	W	47520.92
F	M	13840.04
F	M	15840.38
F	M	22592.02



OUTPUT - (Untitled)

M	M	29700.44
M	M	18750.15
M	M	45650.24
M	M	33675.48
M	M	28975.63
M	M	24500.36
M	M	19200.48
M	M	31900.92
M	M	29200.18
M	M	35200.15
M	M	41900.54

N = 58

USING THE MACRO PROGRAMMING LANGUAGE

The macro programming language consists of statements such as

```
%IF %DO %ELSE %THEN %END
```

among others.

There are also a number of character handling functions such as

```
%UPCASE %SUBSTR %SCAN
```

There are some Call Routines such as

```
CALL SYMPUT
```

For passing values from the data step to a macro variable

We can use these techniques within a macro code bundle to help control the program:

```
PROGRAM EDITOR - (Untitled)
%macro proc(proc,dsetname,var1,var2,var3,agevalue);
  %if %upcase(&proc)=PRINT or %upcase(&proc)=MEANS
  %then %do;
    proc &proc data=&dsetname;
    var &var1 &var2 &var3;
    where age>&agevalue;
    title "Salaries for Employees Over &agevalue years";
    run;
  %end;
  %else %put Please use Proc Print or Proc Means;
%mend proc;

%proc(print,sasuser.demog,gender,status,salary,25)
```

OUTPUT - (Untitled)

Salaries for Employees Over 25 years

OBS	GENDER	STATUS	SALARY
1	F	M	8870.23
3	F	M	28512.66
4	F	M	14840.27
5	F	M	47520.38
6	F	M	23760.44
7	F	M	20592.16
8	F	M	13760.27
9	F	W	47520.92
10	F	M	13840.04
12	F	M	9870.34
15	F	M	15840.38
16	F	M	33520.83

Calling Proc Means with the Macro PROC:

```
PROGRAM EDITOR - (Untitled)

%proc(means,sasuser.demog,age,height,weight,55)
```

OUTPUT - (Untitled)

Salaries for Employees Over 55 years

Variable	N	Mean	Std Dev	Minimum	Maximum
AGE	11	59.8181818	3.3111382	56.0000000	65.0000000
HEIGHT	11	66.2272727	12.7502228	52.0000000	88.5000000
WEIGHT	11	67.6363636	6.1036502	60.0000000	81.0000000



SUMMARY

The Macro Language Facility within the SAS System provides the programmer with more tools with which to control and improve SAS programs. This paper has shown how the 3 main aspects of the macro language can be used to create more generic and therefore more efficient SAS programs. For more information, consult the SAS Macro Language Usage and Reference or contact the author at the address on the next page.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Company: Amadeus Software Limited
Address: Mulberry House, 9 Church Green, Witney, Oxon OX28 4AZ
Work Phone: +44 (0) 1993 848010
Email: info@amadeus.co.uk
Web: www.amadeus.co.uk

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.